> Soluzioni del compito di Calcolo Numerico del 27febbraio 2004

> Enrico Bertolazzi

Esercizio 2

```
> restart:
  with(LinearAlgebra) :
> x:='x';
```

$$x := x$$

Funzione da integrare:

```
> f := x -> sin(2*x)*(1-x) ;
```

$$f := x \rightarrow \sin(2\,x)\,(1 - x)$$

Integrale esatto

```
> a,b := -1,2:
  int(f(x),x=a..b);
  esatto := evalf(%);
```

$$\cos(2) - \frac{1}{4}\sin(2) + \frac{1}{2}\cos(4) - \frac{1}{4}\sin(4)$$

$$esatto := -0.7810923798$$

Derivata quarta della funzione

```
> ddddf := (D@@4)(f) ;
```

$$ddddf := x \rightarrow 16\sin(2\,x)\,(1 - x) + 32\cos(2\,x)$$

Possibile costante che maggiore il modulo della derivata seconda

```
> M := 16*2 + 32 ;
```

$$M := 64$$

Verifica grafica:

```
> plot(abs(ddddf(x)),x=a..b) ;
```



Calcolo in numero minimo stimato di intervalli

```
> err := 10^(-3) = (b-a)^5/(180*n^4)*M ;
```

$$err := \frac{1}{1000} = \frac{432}{5\ n^4}$$

```
> isolate(err,n^4) ;
```

$$n^4 = 86400$$

```
> evalf(rhs(%)^(1/4));
```

$$17.14464258$$

Intervalli stimati: 172

```
> h := (b-a)/18 ;
```

$$h := \frac{1}{6}$$

```
> with(Student[Calculus1]):
> simp := evalf(ApproximateInt(f(x), a..b, method = simpson,partition = 18));
```

simp := -0.7810950826

```
> ApproximateInt(f(x), a..b, method = simpson,partition = 18, output=plot);
```




```
> simp - esatto ;
```

-0.0000027028

Integrale con 6 intervalli

```
> evalf(ApproximateInt(f(x), a..b, method = simpson,partition = 6));
```

-0.7813164831

```
> evalf(int(sin(2*x*Pi/180)*(1-x),x=-1..2));
```

-0.05233966012

Esercizio 3

```
> restart:
  with(LinearAlgebra) :
```

Definizione del problema

```
> p := x -> x ;
  q := x -> -3 ;
  r := x -> -4 ;
  xa, xb := -2, 2 ;
  ya, yb := -3, 3 ;
```

$$p := x \rightarrow x$$

$$q := x \rightarrow -3$$

$$r := x \rightarrow -4$$

$$xa, xb := -2, 2$$

$$ya, yb := -3, 3$$

Differenze finite:

```
> n := 4 ;
  h := (xb-xa) / n ;
  x[0] := xa ;
  x[1] := xa + h ;
  x[2] := xa + 2*h ;
  x[3] := xa + 3*h ;
  x[4] := xa + 4*h ;
```

$$n := 4$$

$$h := 1$$

$$x_0 := -2$$

$$x_1 := -1$$

$$x_2 := 0$$

$$x_3 := 1$$

$$x_4 := 2$$

```
> eq := k -> (y[k+1]-2*y[k]+y[k-1])/h^2 +
            max(p(x[k]),0) * (y[k+1]-y[k])/h +
            min(p(x[k]),0) * (y[k]-y[k-1])/h +
            q(x[k]) * y[k] - r(x[k]) ;
```

$$eq := k \rightarrow \frac{y_{k+1} - 2\, y_k + y_{k-1}}{h^2} + \frac{\max(p(x_k), 0)\,(y_{k+1} - y_k)}{h} + \frac{\min(p(x_k), 0)\,(y_k - y_{k-1})}{h} + q(x_k)\, y_k - r(x_{}$$

Equazioni risultanti

```
> eq1 := eq(1) ;
  eq2 := eq(2) ;
  eq3 := eq(3) ;
```

$$eq1 := y_2 - 6\,y_1 + 2\,y_0 + 4$$
$$eq2 := y_3 - 5\,y_2 + y_1 + 4$$
$$eq3 := 2\,y_4 - 6\,y_3 + y_2 + 4$$

Estraggo il sistema lineare dalle equazioni

```
> A := linalg[genmatrix]([eq1,eq2,eq3],[y[1],y[2],y[3]],'b'):
  A := convert(A,Matrix);
```

$$A := \begin{bmatrix} -6 & 1 & 0 \\ 1 & -5 & 1 \\ 0 & 1 & -6 \end{bmatrix}$$

```
> b := Transpose(convert(b,Vector));
```

$$b := \begin{bmatrix} -2\,y_0 - 4 \\ -4 \\ -2\,y_4 - 4 \end{bmatrix}$$

Sostituisco le condizioni al contorno

```
> b := subs(y[0]=ya, y[4]=yb, b) ;
```

$$b := \begin{bmatrix} 2 \\ -4 \\ -10 \end{bmatrix}$$

Risolvo il sistema lineare

```
> LinearSolve(A,b) ;
```

$$\begin{bmatrix} \frac{-1}{7} \\ \frac{8}{7} \\ \frac{13}{7} \end{bmatrix}$$

Esercizio 4

```
> restart:
  with(LinearAlgebra) :
```

matrice identita e vettori canonici della base usati per i conti successivi

```
> e1,e2,e3,e4 := <1,0,0,0>,<0,1,0,0>,<0,0,1,0>,<0,0,0,1>;
```

$$e1, e2, e3, e4 := \begin{bmatrix}1\\0\\0\\0\end{bmatrix}, \begin{bmatrix}0\\1\\0\\0\end{bmatrix}, \begin{bmatrix}0\\0\\1\\0\end{bmatrix}, \begin{bmatrix}0\\0\\0\\1\end{bmatrix}$$

Matrice identità

```
> ID := <e1|e2|e3|e4>;
```

$$ID := \begin{bmatrix}1 & 0 & 0 & 0\\0 & 1 & 0 & 0\\0 & 0 & 1 & 0\\0 & 0 & 0 & 1\end{bmatrix}$$

Matrice iniziale

```
> A := Matrix([[-1,  0, -1, 4],
              [-1, -2, 4, -1],
              [-1,  4, -1, -2],
              [2,  0, -1, -1]]);
```

$$A := \begin{bmatrix}-1 & 0 & -1 & 4\\-1 & -2 & 4 & -1\\-1 & 4 & -1 & -2\\2 & 0 & -1 & -1\end{bmatrix}$$

```
> A0 := A :
```

Matrice di permutazione

```
> P1 := <e4|e2|e3|e1>;
```

$$P1 := \begin{bmatrix}0 & 0 & 0 & 1\\0 & 1 & 0 & 0\\0 & 0 & 1 & 0\\1 & 0 & 0 & 0\end{bmatrix}$$

Scambio la prima con la quarta riga

```
> P1A0 := P1.A0;
```

$$P1A0 := \begin{bmatrix} 2 & 0 & -1 & -1 \\ -1 & -2 & 4 & -1 \\ -1 & 4 & -1 & -2 \\ -1 & 0 & -1 & 4 \end{bmatrix}$$

Matrice di eliminazione

```
> L1 := ID-<0,P1A0[2,1],P1A0[3,1],P1A0[4,1]>.Transpose(e1)/P1A0[1,1];
```

$$L1 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ \frac{1}{2} & 1 & 0 & 0 \\ \frac{1}{2} & 0 & 1 & 0 \\ \frac{1}{2} & 0 & 0 & 1 \end{bmatrix}$$

Primo passo del metodo di Gauss

```
> A1 := L1.P1A0;
```

$$A1 := \begin{bmatrix} 2 & 0 & -1 & -1 \\ 0 & -2 & \frac{7}{2} & \frac{-3}{2} \\ 0 & 4 & \frac{-3}{2} & \frac{-5}{2} \\ 0 & 0 & \frac{-3}{2} & \frac{7}{2} \end{bmatrix}$$

Matrice di scambio seconda e terza riga

```
> P2 := <e1|e3|e2|e4>;
```

$$P2 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Scambio la seconda con la terza riga

```
> P2A1 := P2.A1;
```

$$P2A1 := \begin{bmatrix} 2 & 0 & -1 & -1 \\ 0 & 4 & \frac{-3}{2} & \frac{-5}{2} \\ 0 & -2 & \frac{7}{2} & \frac{-3}{2} \\ 0 & 0 & \frac{-3}{2} & \frac{7}{2} \end{bmatrix}$$

Matrice di eliminazione

```
> L2 := ID-<0,0,P2A1[3,2],P2A1[4,2]>.Transpose(e2)/P2A1[2,2];
```

$$L2 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & \frac{1}{2} & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Secondo Passo del metodo di Gauss

```
> A2 := L2.P2A1;
```

$$A2 := \begin{bmatrix} 2 & 0 & -1 & -1 \\ 0 & 4 & \frac{-3}{2} & \frac{-5}{2} \\ 0 & 0 & \frac{11}{4} & \frac{-11}{4} \\ 0 & 0 & \frac{-3}{2} & \frac{7}{2} \end{bmatrix}$$

Nessuno scambio

```
> P3 := <e1|e2|e3|e4>;
```

$$P3 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Nessuno Scambio

```
> P3A2 := P3.A2;
```

$$P3A2 := \begin{bmatrix} 2 & 0 & -1 & -1 \\ 0 & 4 & \frac{-3}{2} & \frac{-5}{2} \\ 0 & 0 & \frac{11}{4} & \frac{-11}{4} \\ 0 & 0 & \frac{-3}{2} & \frac{7}{2} \end{bmatrix}$$

Matrice di eliminazione

```
> L3 := ID-<0,0,0,P3A2[4,3]>.Transpose(e3)/P3A2[3,3];
```

$$L3 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & \frac{6}{11} & 1 \end{bmatrix}$$

Terzo passo del metodo di Gauss

```
> A3 := L3.P3A2;
```

$$A3 := \begin{bmatrix} 2 & 0 & -1 & -1 \\ 0 & 4 & \frac{-3}{2} & \frac{-5}{2} \\ 0 & 0 & \frac{11}{4} & \frac{-11}{4} \\ 0 & 0 & 0 & 2 \end{bmatrix}$$

Matrici P, L, U

```
> P := P3.P2.P1:
  L := P.(L3.P3.L2.P2.L1.P1)^(-1):
  U := A3:
  P, L, U ;
```

$$\begin{bmatrix} 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 & 0 \\ \frac{-1}{2} & 1 & 0 & 0 \\ \frac{-1}{2} & \frac{-1}{2} & 1 & 0 \\ \frac{-1}{2} & 0 & \frac{-6}{11} & 1 \end{bmatrix}, \begin{bmatrix} 2 & 0 & -1 & -1 \\ 0 & 4 & \frac{-3}{2} & \frac{-5}{2} \\ 0 & 0 & \frac{11}{4} & \frac{-11}{4} \\ 0 & 0 & 0 & 2 \end{bmatrix}$$

Controllo

```
> R := P.A - L.U;
```

$$R := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

Calcolo la soluzione del primo problema

```
> b  := A.< -1, 2, 2, 1>:
  Pb := P.b:
  z  := L^(-1).Pb:
  x  := U^(-1).z:
  b, Pb, z, x ;
```

$$\begin{bmatrix} 3 \\ 4 \\ 5 \\ -5 \end{bmatrix}, \begin{bmatrix} -5 \\ 5 \\ 4 \\ 3 \end{bmatrix}, \begin{bmatrix} -5 \\ \frac{5}{2} \\ \frac{11}{4} \\ 2 \end{bmatrix}, \begin{bmatrix} -1 \\ 2 \\ 2 \\ 1 \end{bmatrix}$$

Calcolo la soluzione del secondo problema

```
> b  := A.< 1, 1, 0, 0>:
  Pb := P.b:
  z  := L^(-1).Pb:
  x  := U^(-1).z:
  b, Pb, z, x ;
```

$$\begin{bmatrix} -1 \\ -3 \\ 3 \\ 2 \end{bmatrix}, \begin{bmatrix} 2 \\ 3 \\ -3 \\ -1 \end{bmatrix}, \begin{bmatrix} 2 \\ 4 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 1 \\ 1 \\ 0 \\ 0 \end{bmatrix}$$

Esercizio 5

```
> restart:
  with(LinearAlgebra) :
```

Funzione da integrare

```
> f := (x,y) -> x^2 - 2*x*y ;
```

$$f := (x, y) \rightarrow x^2 - 2\,x\,y$$

Derivata prima

```
> dy := f(x,y) ;
```

$$dy := x^2 - 2\,x\,y$$

Derivata seconda

```
> ddy := diff(dy,x)+diff(dy,y)*f(x,y) ;
  expand(ddy) ;
```

$$ddy := 2\,x - 2\,y - 2\,x\,(x^2 - 2\,x\,y)$$

$$2\,x - 2\,y - 2\,x^3 + 4\,x^2\,y$$

Derivata terza

```
> dddy := diff(ddy,x)+diff(ddy,y)*f(x,y) ;
  expand(dddy) ;
```

$$dddy := 2 - 2\,x^2 + 4\,x\,y - 2\,x\,(2\,x - 2\,y) + (-2 + 4\,x^2)\,(x^2 - 2\,x\,y)$$

$$2 - 8\,x^2 + 12\,x\,y + 4\,x^4 - 8\,x^3\,y$$

Costruzione della serie di Taylor troncata

```
> ynew := unapply( y+dy*h+ddy*h^2/2+dddy*h^3/6, x, y, h) ;
```

$$ynew := (x, y, h) \rightarrow y + (x^2 - 2\,x\,y)\,h + \frac{1}{2}\,(2\,x - 2\,y - 2\,x\,(x^2 - 2\,x\,y))\,h^2$$
$$+ \frac{1}{6}\,(2 - 2\,x^2 + 4\,x\,y - 2\,x\,(2\,x - 2\,y) + (-2 + 4\,x^2)\,(x^2 - 2\,x\,y))\,h^3$$

Calcolo alcuni passi

```
> x0, y0 := 1, 2;
```

$$x0, y0 := 1, 2$$

```
> h := 0.2 ;
```

$h := 0.2$

```
> x1 := x0 + h ;
  y1 := ynew(x0, y0, h) ;
```

$x1 := 1.2$

$y1 := 1.488000000$

```
> x2 := x1 + h ;
  y2 := ynew(x1, y1, h) ;
```

$x2 := 1.4$

$y2 := 1.152046251$

```
> x3 := x2 + h ;
  y3 := ynew(x2, y2, h) ;
```

$x3 := 1.6$

$y3 := 0.9740340963$

Esercizio 6

```
> restart:
  with(LinearAlgebra) :
```

```
> # Punti di interpolazione
```

```
> X0,X1,X2,X3,X4 := 0,1,2,3,4 ;
  Y0,Y1,Y2,Y3,Y4 := 1,2,9,28,65 ;
```

$X0, X1, X2, X3, X4 := 0, 1, 2, 3, 4$

$Y0, Y1, Y2, Y3, Y4 := 1, 2, 9, 28, 65$

```
> # condizioni al contorno
  S2_init  := 0 ;
  S2_final := 24 ;
```

$S2\_init := 0$

$S2\_final := 24$

```
> # Equazioni dei momenti
  # notate che h[k] = 1 costante perche i punti sono
  # equispaziati
  EQ1     := M0*(1/2) + 2*M1 + M2*(1/2) = 3*((Y2-Y1)-(Y1-Y0)) ;
  EQ2     := M1*(1/2) + 2*M2 + M3*(1/2) = 3*((Y3-Y2)-(Y2-Y1)) ;
  EQ3     := M2*(1/2) + 2*M3 + M4*(1/2) = 3*((Y4-Y3)-(Y3-Y2)) ;
  EQINIT  := M0 = S2_init ;
  EQFINAL := M4 = S2_final ;
```

$$EQ1 := \frac{1}{2} M0 + 2\, M1 + \frac{1}{2} M2 = 18$$

$$EQ2 := \frac{1}{2} M1 + 2\, M2 + \frac{1}{2} M3 = 36$$

$$EQ3 := \frac{1}{2}\,M2 + 2\,M3 + \frac{1}{2}\,M4 = 54$$

$$EQINIT := M0 = 0$$

$$EQFINAL := M4 = 24$$

```
> RES := solve({EQ1,EQ2,EQ3,EQINIT,EQFINAL},{M0,M1,M2,M3,M4}) ;
```

$$RES := \{M3 = 18, M1 = 6, M0 = 0, M4 = 24, M2 = 12\}$$

```
> # Calcolo i tratti di cubica
  S1 := Y0 + (Y1-Y0 - (M1+2*M0) / 6) * (x-X0)
           + (M0/2)*(x-X0)^2
           + ((M1-M0)/6)*(x-X0)^3 ;
  subs(RES,S1) ;
```

$$S1 := 1 + \left(1 - \frac{1}{6}\,M1 - \frac{1}{3}\,M0\right)\,x + \frac{1}{2}\,M0\,x^2 + \frac{1}{6}\,(M1 - M0)\,x^3$$

$$1 + x^3$$

```
> S2 := Y1 + (Y2-Y1 - (M2+2*M1) / 6) * (x-X1)
           + (M1/2)*(x-X1)^2
           + ((M2-M1)/6)*(x-X1)^3 ;
  expand(subs(RES,S2)) ;
```

$$S2 := 2 + \left(7 - \frac{1}{6}\,M2 - \frac{1}{3}\,M1\right)\,(x - 1) + \frac{1}{2}\,M1\,(x - 1)^2 + \frac{1}{6}\,(M2 - M1)\,(x - 1)^3$$

$$1 + x^3$$

```
> S3 := Y2 + (Y3-Y2 - (M3+2*M2) / 6) * (x-X2)
           + (M2/2)*(x-X2)^2
           + ((M3-M2)/6)*(x-X2)^3 ;
  expand(subs(RES,S3)) ;
```

$$S3 := 9 + \left(19 - \frac{1}{6}\,M3 - \frac{1}{3}\,M2\right)\,(x - 2) + \frac{1}{2}\,M2\,(x - 2)^2 + \frac{1}{6}\,(M3 - M2)\,(x - 2)^3$$

$$1 + x^3$$

```
> S4 := Y3 + (Y4-Y3 - (M4+2*M3) / 6) * (x-X3)
           + (M3/2)*(x-X3)^2
           + ((M4-M3)/6)*(x-X3)^3 ;
  expand(subs(RES,S3)) ;
```

$$S4 := 28 + \left(37 - \frac{1}{6}\,M4 - \frac{1}{3}\,M3\right)\,(x - 3) + \frac{1}{2}\,M3\,(x - 3)^2 + \frac{1}{6}\,(M4 - M3)\,(x - 3)^3$$

$$1 + x^3$$

```
>
```